



T-Swap Protocol Audit Report

Version 1.0

Lulox

July 2, 2024

T-Swap Protocol Audit Report

Lulox

July 02, 2024

Prepared by: Lulox Lead Auditors:

- Lulox

Table of Contents

- Table of Contents
- Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
 - Roles
- Executive Summary
 - Issues found
- Findings
- High
- Medium
- Low
- Informational
- Gas

Protocol Summary

This project is meant to be a permissionless way for users to swap assets between each other at a fair price. You can think of T-Swap as a decentralized asset/token exchange (DEX). T-Swap is known as an Automated Market Maker (AMM) because it doesn't use a normal "order book" style exchange, instead it uses "Pools" of an asset. It is similar to Uniswap. To understand Uniswap, please watch this video: [Uniswap Explained](#)

The core invariant of the protocol is: $x * y = k$. In practice though, the protocol takes fees and actually increases k . So we need to make sure $x * y = k$ before fees are applied.

Disclaimer

The Lulox team makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by the team is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

Audit Details

The findings described in this document correspond the following commit hash: [8803f851f6b37e99eab2e94b4690c8b70e26b3f6](#)

Scope

```
1 ./src/  
2 #-- PoolFactory.sol  
3 #-- TSwapPool.sol
```

Solc Version: 0.8.20 Chain(s) to deploy contract to: Ethereum Tokens: Any ERC20 token

Roles

- Liquidity Providers: Users who have liquidity deposited into the pools. Their shares are represented by the LP ERC20 tokens. They gain a 0.3% fee every time a swap is made.
- Users: Users who want to swap tokens

Executive Summary

Issues found

Severity	Number of issues found
High	5
Medium	3
Low	3
Info	5
Gas	4
Total	20

Findings

High

[H-1] Incorrect fee calculation in TSwapPool::getInputAmountBasedOnOutput causes TSwapPool::swapExactOutput to take a ~90% fee from users

Description: The `getInputAmountBasedOnOutput` function is intended to calculate the amount of tokens a user should send in exchange of a specific amount of output tokens. However, the function

currently miscalculates the resulting amount. When calculating the fee, it scales the amount by 10000 instead of 1000 required to calculate the 0.3% fee specified in the project's [README.md](#).

It reads: "The TSwap protocol accrues fees from users who make swaps. **Every swap has a 0.3 fee**, represented in `getInputAmountBasedOnOutput` and `getOutputAmountBasedOnInput`. **Each applies a 997 out of 1000 multiplier**. That fee stays in the protocol."

Impact: Protocol takes more fees than expected from users.

Correct calculation of the fee should be:

997 out of 1000

$997 / 1000 = 0.997$ | Multiplier

$0.997 * 100 = 99.7\%$ | Total after fee

$100\% - 99.7\% = 0.3\%$ | Fee

However, `getInputAmountBasedOnOutput` currently calculates the fee as:

997 out of 10000

$997 / 10000 = 0.0997$ | Multiplier

$0.0997 * 100 = 9.97\%$ | Total after fee

$100\% - 9.97\% = 90.03\%$ | Fee

Proof of Concept: Every time a user swaps with `TSwapPool.swapExactOutput`, fee is miscalculated.

Include this test in `TSwapPool.t.sol`:

Proof of Code

```
1  function testGetInputAmountBasedOnOutputIncorrectFeeCalculation()
2      public {
3      // Liquidity provider adds liquidity to the pool
4      testDeposit();
5
6      // User starts out with 10e18 pool tokens, but only intends to
7      // spend one
8      uint256 intendedInput = 1e18;
9      uint256 userPoolTokenBalanceBefore = poolToken.balanceOf(
10         address(user));
11     console.log("User pool token balance before: ",
12         userPoolTokenBalanceBefore);
13     console.log("Intended input: ", intendedInput);
14     // We calculate the expected output amount based on the input
15     // amount we're willing to spend
16     uint256 expectedOutput = pool.getOutputAmountBasedOnInput(
```

```
12         intendedInput, poolToken.balanceOf(address(pool)), weth.
           balanceOf(address(pool))
13     );
14
15     vm.startPrank(user);
16     // User approves more tokens than its willing to swap
17     poolToken.approve(address(pool), type(uint256).max);
18     // swapExactOutput returns the actual input amount incorrectly
       given by getInputAmountBasedOnOutput
19     uint256 actualInput = pool.swapExactOutput(poolToken, weth,
       expectedOutput, uint64(block.timestamp));
20     uint256 userPoolTokenBalanceAfter = poolToken.balanceOf(address
       (user));
21     console.log("Actual input: ", actualInput);
22     console.log("User pool token balance after: ",
       userPoolTokenBalanceAfter);
23
24     // The intended input amount is different from the actual input
       amount
25     assert(actualInput > intendedInput);
26     assert(userPoolTokenBalanceBefore - userPoolTokenBalanceAfter >
       intendedInput);
27     // The swap drained more than 90% of the user's pool tokens
28     assert(userPoolTokenBalanceAfter < 1e18);
29 }
```

Recommended Mitigation:

Don't use magic numbers, declare them as constants instead.

```
1 +     uint256 private constant PRECISION = 1000;
2 +     uint256 private constant FEE = 997;
3 .
4 .
5 .
6     function getInputAmountBasedOnOutput(
7         uint256 outputAmount,
8         uint256 inputReserves,
9         uint256 outputReserves
10    )
11    public
12    pure
13    revertIfZero(outputAmount)
14    revertIfZero(outputReserves)
15    returns (uint256 inputAmount)
16    {
17 -         return ((inputReserves * outputAmount) * 10000) / ((
           outputReserves - outputAmount) * 997);
18 +         return ((inputReserves * outputAmount) * PRECISION) / ((
           outputReserves - outputAmount) * FEE);
19    }
```

[H-2] Lack of slippage protection in TSwapPool::swapExactOutput exposes user to MEV and causes to spend more than expected

Description: The `TSwapPool::swapExactOutput` function does not include any sort of slippage protection. This function is similar to what is done in `TSwapPool::swapExactInput`, where the function specifies a `minOutputAmount`, the `swapExactOutput` function should specify a `maxInputAmount` to limit the amount of tokens the user is willing to pay.

Impact: This vulnerability makes the user susceptible to MEV (sandwich attacks and frontrunning attacks).

Any user on the Ethereum network has the ability to watch for new transactions being sent to the network. When the attacker sees a large victim transaction that they want to front run come in, they can create a similar transaction that would move the market up. They then increase their gas fees to ensure that their order gets executed first. The attacker transaction executes, raising the price of the asset, and then the victim transaction executes at the higher price. The attacker is then free to exit the position immediately, pocketing the difference, having never exposed themselves to any risk.

Sophisticated front-runners will likely call these transactions from their own contract addresses to make sure they end up with the prices they expect, and don't collide with other front-runners.

For example:

1. The price of 1 WETH right now is 1000 USDC
2. User inputs a `swapExactOutput` looking for 1 WETH
 1. `inputToken = USDC`
 2. `outputToken = WETH`
 3. `outputAmount = 1`
 4. `deadline = whatever`
3. The function does not offer a `maxInputAmount`
4. As the transaction is pending in the mempool, the market changes! And the price moves HUGE -> 1 WETH is now 10000 USDC. 10x more than the user expected.
5. The transaction completes, but the user sent the protocol 10000 USDC instead of the expected 1000 USDC.

Proof of Concept:

Proof of Code

Add this test to `TSwapPool.t.sol`:

```
1 function testSwapExactOutputLackOfSlippageProtection() public {
2     vm.startPrank(LiquidityProvider);
```

```
3      weth.approve(address(pool), 100e18);
4      poolToken.approve(address(pool), 100e18);
5      pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6      vm.stopPrank();
7
8      // This is what the user estimates that will receive after the
      swap
9      uint256 OUTPUT_WETH = 1e17;
10     uint256 expectedPoolTokenToSpend = pool.
        getInputAmountBasedOnOutput(
11         OUTPUT_WETH, poolToken.balanceOf(address(pool)), weth.
            balanceOf(address(pool))
12     );
13
14     // mevAttacker sees the transaction of the user and puts this
        before on the mempool
15     address mevAttacker = makeAddr("mevAttacker");
16     vm.startPrank(mevAttacker);
17     // We assume the attack comes from a whale or after a flashloan
18     uint256 MEV_POOL_TOKENS = 50e18;
19     poolToken.mint(mevAttacker, MEV_POOL_TOKENS);
20     poolToken.approve(address(pool), type(uint256).max);
21     pool.swapExactInput(poolToken, MEV_POOL_TOKENS, weth, 0, uint64(
        block.timestamp));
22     vm.stopPrank();
23
24     // User transaction goes through
25     vm.startPrank(user);
26     poolToken.approve(address(pool), type(uint256).max);
27     poolToken.mint(user, 100e18);
28     uint256 poolTokensSpentByUser = pool.swapExactOutput(poolToken,
        weth, OUTPUT_WETH, uint64(block.timestamp));
29     vm.stopPrank();
30
31     // mevAttacker completes the sandwich and takes profits
32     vm.startPrank(mevAttacker);
33     weth.approve(address(pool), type(uint256).max);
34     uint256 wethBalanceOfMevAttacker = weth.balanceOf(address(
        mevAttacker));
35     pool.swapExactInput(weth, wethBalanceOfMevAttacker, poolToken,
        0, uint64(block.timestamp));
36     vm.stopPrank();
37
38     uint256 poolTokenBalanceOfMevAttackerAfterSandwich = poolToken.
        balanceOf(address(mevAttacker));
39
40     assert(poolTokenBalanceOfMevAttackerAfterSandwich >
        MEV_POOL_TOKENS);
41     assert(poolTokensSpentByUser > expectedPoolTokenToSpend);
42 }
```


Recommended Mitigation: Include a `maxInputAmount` so the user only has to spend up to a specific amount, and can predict how much they will spend on the protocol.

```
1  function swapExactOutput(  
2      IERC20 inputToken,  
3      IERC20 outputToken,  
4      uint256 outputAmount,  
5 +     uint256 maxInputAmount  
6      uint64 deadline  
7  )  
8      public  
9      revertIfZero(outputAmount)  
10     revertIfDeadlinePassed(deadline)  
11     returns (uint256 inputAmount)  
12  {  
13     uint256 inputReserves = inputToken.balanceOf(address(this));  
14     uint256 outputReserves = outputToken.balanceOf(address(this));  
15  
16 +     if (inputAmount > maxInputAmount) {  
17 +         revert TSwapPool__InputTooHigh(inputAmount, maxInputAmount  
18 +     );  
19     }  
20     inputAmount = getInputAmountBasedOnOutput(outputAmount,  
21         inputReserves, outputReserves);  
22     _swap(inputToken, inputAmount, outputToken, outputAmount);  
23 }
```

[H-3] TSwapPool::sellPoolTokens mismatches input and output tokens causing users to spend more than intended

Description: The `TSwapPool::sellPoolTokens` function is intended to allow users to easily sell pool tokens and receive WETH in exchange. Users indicate how many pool tokens they're willing to sell in the `poolTokenAmount` parameter. However, the function currently miscalculates the swapped amount, making the user spend more than intended to get the exact amount of WETH specified in the parameter.

Impact: Users will swap the wrong amount of tokens, which breaks the protocol functionality.

Proof of Concept:

```
1  function testSellPoolTokensWorksBackwards() public {  
2      vm.startPrank(liquidityProvider);  
3      weth.approve(address(pool), 100e18);  
4      poolToken.approve(address(pool), 100e18);  
5      pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
```

```

6      vm.stopPrank();
7
8      uint256 USER_POOL_TOKENS = 1e18;
9      poolToken.mint(user, USER_POOL_TOKENS);
10     uint256 balanceWethBefore = weth.balanceOf(address(user));
11     uint256 balancePoolTokenBefore = poolToken.balanceOf(address(
        user));
12
13     vm.startPrank(user);
14     poolToken.approve(address(pool), type(uint256).max);
15     pool.sellPoolTokens(USER_POOL_TOKENS);
16
17     uint256 balanceWethAfter = weth.balanceOf(address(user));
18     uint256 balancePoolTokenAfter = poolToken.balanceOf(address(
        user));
19     // Here we assert that the user bought the amount of WETH
        specified in the parameter of sellPoolTokens
20     assert(balanceWethAfter - balanceWethBefore == USER_POOL_TOKENS
        );
21     // And here we assert the sellPoolTokens makes the user spend
        more than intended in the badly named parameter
22     assert(balancePoolTokenBefore - balancePoolTokenAfter >
        USER_POOL_TOKENS);
23 }

```

Recommended Mitigation: Consider changing the implementation to use `swapExactInput` instead of `swapExactOutput`. Note that this would also require changing the `sellPoolTokens` function to accept a new parameter (ie `minWethToReceive` to be passed to `swapExactInput`).

```

1  function sellPoolTokens(
2      uint256 poolTokenAmount,
3  +   uint256 minWethToReceive
4  ) external returns (uint256 wethAmount) {
5  -     return swapExactOutput(i_poolToken, i_wethToken,
        poolTokenAmount, uint64(block.timestamp));
6  +     return swapExactInput(i_poolToken, poolTokenAmount,
        i_wethToken, minWethToReceive, uint64(block.timestamp));
7  }

```

Additionally, it might be wise to add a deadline to the function, as there is currently no deadline. MEV later

[H-4] TSwapPool : : _swap incentive breaks the protocol invariant of $x * y = k$ and allows draining the contract

Description: The following block of code is responsible for the issue:

```

1  swap_count++;

```

```
2  if (swap_count >= SWAP_COUNT_MAX) {
3      swap_count = 0;
4      outputToken.safeTransfer(msg.sender, 1_000_000_000_000_000_000);
5  }
```

An attacker could drain the tokens from both pools by getting this incentive with low value transactions.

Also, the protocol follows a strict invariant of $x * y = k$ (plus fees). Where:

- x is the amount of the pool token
- y is the amount of the WETH token
- k is the constant product of the two balances

This means that whenever the balances change in the protocol, the ratio between the two amounts should remain constant, hence the k . However, this is broken due to the extra incentive in the `_swap` function. Meaning that over time, the protocol funds will be drained.

The following block of code is responsible for the issue:

```
1  swap_count++;
2  if (swap_count >= SWAP_COUNT_MAX) {
3      swap_count = 0;
4      outputToken.safeTransfer(msg.sender, 1_000_000_000_000_000_000);
5  }
```

Impact: A user could maliciously drain the protocol of funds by doing a lot of swaps and collecting the extra incentive given out by the protocol.

Most simply put, the protocol's core invariant is broken.

Proof of Concept:

1. A user swaps 10 times, and collects the extra incentive of 1_000_000_000_000_000_000 tokens
2. The user continues to swap until all the protocol funds are drained

Draining contract Proof of Code

```
1  function testSwapCountAllowsDrainingTheContract() public {
2      vm.startPrank(LiquidityProvider);
3      weth.approve(address(pool), 10e18);
4      poolToken.approve(address(pool), 10e18);
5      pool.deposit(10e18, 10e18, 10e18, uint64(block.timestamp));
6      vm.stopPrank();
7
8      address attacker = makeAddr("attacker");
9      weth.mint(attacker, 1e18);
```

```
10     poolToken.mint(attacker, 1e18);
11
12     vm.startPrank(attacker);
13     poolToken.approve(address(pool), 1e18);
14     weth.approve(address(pool), 1e18);
15     for (uint256 i = 0; i < 90; i++) {
16         pool.swapExactInput(poolToken, 1e15, weth, 0, uint64(block.
            timestamp));
17     }
18     for (uint256 i = 0; i < 90; i++) {
19         pool.swapExactInput(weth, 1e15, poolToken, 0, uint64(block.
            timestamp));
20     }
21     vm.stopPrank();
22
23     // Attacker ends with more than intended, and pool gets drained
        of its funds
24     assert(poolToken.balanceOf(attacker) > 9e18);
25     assert(weth.balanceOf(attacker) > 9e18);
26     assert(weth.balanceOf(address(pool)) < 2e18);
27     assert(poolToken.balanceOf(address(pool)) < 2e18);
28 }
```

Invariant Break Proof of Code

```
1     function testInvariantBreak() public {
2         vm.startPrank(liquidityProvider);
3         weth.approve(address(pool), 100e18);
4         poolToken.approve(address(pool), 100e18);
5         pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6         vm.stopPrank();
7
8         vm.startPrank(user);
9         poolToken.approve(address(pool), type(uint256).max);
10        poolToken.mint(user, 100e18);
11        // We call swap 9 times to increase the swap_count
12        uint256 outputWeth = 1e17;
13        for (uint256 i = 0; i < 9; i++) {
14            pool.swapExactOutput(poolToken, weth, outputWeth, uint64(
                block.timestamp));
15        }
16
17        int256 startingY = int256(weth.balanceOf(address(pool)));
18        int256 expectedDeltaY = int256(-1) * int256(outputWeth);
19        // The 10th swap gets the incentive and breaks the invariant
20        pool.swapExactOutput(poolToken, weth, outputWeth, uint64(block.
            timestamp));
21        vm.stopPrank();
22
23        uint256 endingY = weth.balanceOf(address(pool));
24        int256 actualDeltaY = int256(endingY) - int256(startingY);
```

```
25
26     assertEq(actualDeltaY, expectedDeltaY);
27 }
```

Recommended Mitigation: Remove the extra incentive mechanism.

```
1 -     uint256 private swap_count = 0;
2 -     uint256 private constant SWAP_COUNT_MAX = 10;
3 -
4 -
5 -
6 -     swap_count++;
7 -     if (swap_count >= SWAP_COUNT_MAX) {
8 -         swap_count = 0;
9 -         outputToken.safeTransfer(msg.sender, 1
10 -             _000_000_000_000_000_000);
10 -     }
```

[H-5] Missing slippage protection in TSwapPool::sellPoolTokens allows MEV searchers to profit from user

Relevant GitHub links:

TSwapPool.sol#L363

Description:

The transaction can be front-run or back-run by a MEV searcher, who can manipulate the order of transactions in a block to their advantage, thus extracting value from the user's transaction. This can be done by reordering transactions in a block to maximize the MEV searcher's profit.

Also, sellPoolTokens has a parameter called poolTokenAmount, which is the amount of pool tokens the user wants to sell. However, this is misleading, as the function actually calls swapExactOutput, and thus the parameter calculates the amount of WETH the user expects to receive. This can lead to the user spending more pool tokens than expected, either calculated or intended.

Impact:

This vulnerability makes the user susceptible to MEV (sandwich attacks and frontrunning attacks).

Any user on the Ethereum network has the ability to watch for new transactions being sent to the network. When the attacker sees a large victim transaction that they want to front run come in, they can create a similar transaction that would move the market up. They then increase their gas fees to ensure that their order gets executed first. The attacker transaction executes, raising the price of the asset, and then the victim transaction executes at the higher price. The attacker is then free to exit the position immediately, pocketing the difference, having never exposed themselves to any risk.

Sophisticated front-runners will likely call these transactions from their own contract addresses to make sure they end up with the prices they expect, and don't collide with other front-runners.

Proof of Concept:

Include this test in `TSwapPool.t.sol`:

Proof of Code

```
1      function testSellPoolTokensLackOfSlippageProtection() public {
2          vm.startPrank(LiquidityProvider);
3          weth.approve(address(pool), 100e18);
4          poolToken.approve(address(pool), 100e18);
5          pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6          vm.stopPrank();
7
8          uint256 userPoolTokenStartingBalance = poolToken.balanceOf(
9              address(user));
10         uint256 userWethStartingBalance = weth.balanceOf(address(user))
11             ;
12         // This is what the user estimates that will receive after
13         // calling the function
14         uint256 wethToReceive = 1e18;
15         // Because of how sellPoolTokens is written, the
16         // userPoolTokensToSell is actually the WETH amount to receive
17         uint256 expectedPoolTokensToSpend = pool.
18             getInputAmountBasedOnOutput(
19                 wethToReceive, poolToken.balanceOf(address(pool)), weth.
20                 balanceOf(address(pool))
21             );
22
23         // mevAttacker sees the transaction of the user and puts this
24         // before on the mempool
25         address mevAttacker = makeAddr("mevAttacker");
26         vm.startPrank(mevAttacker);
27         // We assume the attack comes from a whale or after a flashloan
28         uint256 mevAttackerPoolTokenBalanceBeforeSandwich = 50e18;
29         poolToken.mint(mevAttacker,
30             mevAttackerPoolTokenBalanceBeforeSandwich);
31         poolToken.approve(address(pool), type(uint256).max);
32         pool.swapExactInput(poolToken,
33             mevAttackerPoolTokenBalanceBeforeSandwich, weth, 0, uint64(
34                 block.timestamp));
35         vm.stopPrank();
36
37         // User transaction goes through
38         vm.startPrank(user);
39         poolToken.approve(address(pool), type(uint256).max);
40         poolToken.mint(user, 100e18);
41         uint256 actualWethReceived = pool.sellPoolTokens(wethToReceive)
42             ;
```

```
32     vm.stopPrank();
33     uint256 userPoolTokenEndingBalance = poolToken.balanceOf(
34         address(user));
35     uint256 userWethEndingBalance = weth.balanceOf(address(user));
36
37     // mevAttacker completes the sandwich and takes profits
38     vm.startPrank(mevAttacker);
39     weth.approve(address(pool), type(uint256).max);
40     pool.swapExactInput(weth, weth.balanceOf(address(mevAttacker)),
41         poolToken, 0, uint64(block.timestamp));
42     vm.stopPrank();
43
44     // MEV attacker has more pool tokens than before the sandwich
45     uint256 mevAttackerPoolTokenBalanceAfterSandwich = poolToken.
46         balanceOf(address(mevAttacker));
47     assert(mevAttackerPoolTokenBalanceAfterSandwich >
48         mevAttackerPoolTokenBalanceBeforeSandwich);
49
50     // Remember wethToReceive is actually the parameter
51     poolTokenAmount in sellPoolTokens
52     // User spends more pool tokens than expected, either
53     calculated or intended
54     assert(expectedPoolTokensToSpend < userPoolTokenEndingBalance -
55         userPoolTokenStartingBalance);
56     assert(wethToReceive < userPoolTokenEndingBalance -
57         userPoolTokenStartingBalance);
58     // sellPoolTokens returns more WETH than actually received by
59     the user
60     assert(actualWethReceived > userWethEndingBalance -
61         userWethStartingBalance);
62     // It's also more than the value specified in the (unintended)
63     parameter of sellPoolTokens
64     assert(actualWethReceived > wethToReceive);
65 }
```

Recommended Mitigation:

1. Allow users to specify a slippage tolerance. This protects the user from executing a transaction in unfavorable conditions. Using `TSwapPool::swapExactInput` within `sellPoolTokens` instead of `TSwapPool::swapExactOutput` (which currently has no slippage protection) would allow the user to specify a `minOutputAmount` to set a floor in the amount of tokens the user is expecting to receive. This would protect the user from receiving fewer tokens than expected.

```
1     function sellPoolTokens(
2         uint256 poolTokenAmount,
3         uint64 deadline
4     +     uint256 minWethToReceive
5     ) external returns (uint256 wethAmount) {
```

```
6 -     return swapExactOutput(i_poolToken, i_wethToken,
7 +     poolTokenAmount, deadline);
8 +     return swapExactInput(i_poolToken, poolTokenAmount,
9 +     i_wethToken, minWethToReceive, deadline);
10 }
```

2. Use Flashbots Protect to avoid MEV attacks.

Medium

[M-1] TSwapPool::deposit is missing deadline check, causing transactions to complete even after the deadline

Relevant GitHub links:

TSwapPool.sol#L117

Description: The deposit function accepts a deadline parameter, which according to the natspec is “The deadline for the transaction to be completed by”. However, this parameter is never used. As a consequence, operations that add liquidity to the pool might be executed at unexpected times, in market conditions where the deposit rate is unfavorable.

Impact: The `deadline` parameter is unused. Transactions could be sent when market conditions are unfavorable to deposit, even when adding a deadline parameter.

Proof of Concept:

Include this test in `TSwapPool.t.sol`:

```
1  function testDepositIsMissingADeadlineCheck() public {
2      // Liquidity provider adds liquidity to the pool
3      testDeposit();
4
5      // The user specifies a max deadline for the deposit to happen
6      uint64 deadline = uint64(block.timestamp + 5 minutes);
7      // The transaction stays pending longer than specified in the
8      // deadline parameter and goes through
9      vm.warp(block.timestamp + 30 minutes);
10     // We assert that the current block.timestamp is greater than
11     // the deadline
12     assert(block.timestamp > deadline);
13
14     vm.startPrank(user);
15     poolToken.approve(address(pool), 1e18);
16     weth.approve(address(pool), 1e18);
17     // The user tries to deposit with a deadline in the past
18     uint256 liquidityTokensMinted = pool.deposit(1e18, 1e18, 1e18,
19     deadline);
```



```
17
18     // Deposit goes through even though the deadline has passed,
19     // and liquidity tokens are minted
20     assert(liquidityTokensMinted > 0);
21 }
```

Recommended Mitigation: Consider making the following change to the function.

```
1     function deposit(
2         uint256 wethToDeposit,
3         uint256 minimumLiquidityTokensToMint,
4         uint256 maximumPoolTokensToDeposit,
5         uint64 deadline
6     )
7     external
8     revertIfZero(wethToDeposit)
9     + revertIfDeadlinePassed(deadline)
10    returns (uint256 liquidityTokensToMint)
```

[M-2] Missing deadline parameter in TSwapPool::sellPoolTokens allows pending transactions to be excluded

Relevant GitHub links:

TSwapPool.sol#L363

Description:

In Ethereum, miners have some control over the timestamp of the blocks they mine. This flexibility allows them to include or exclude certain transactions in a block and manipulate the timestamp within a certain range. Miners can slightly adjust the block.timestamp to make certain transactions valid or invalid depending on their preference. For example, if the deadline is set to block.timestamp, a miner could manipulate the timestamp to expire a transaction that they don't want to include or to profit from it in some other way.

Impact:

Using block.timestamp as the deadline parameter in a swap function on a Uniswap V1 clone can introduce a risk of Miner Extractable Value (MEV).

Recommended Mitigation:

1. Add a deadline parameter to the `sellPoolTokens` function to protect users from MEV attacks.

```
1     function sellPoolTokens(
2         uint256 poolTokenAmount
3     +     uint64 deadline
```

```
4     ) external
5 +   revertIfDeadlinePassed(deadline)
6   returns (uint256 wethAmount) {
7       return swapExactOutput(i_poolToken, i_wethToken,
8                               poolTokenAmount, deadline);
9   }
```

2. Use Flashbots Protect to call the transactions to avoid MEV attacks.

[M-3] Rebase, fee-on-transfer, and ERC777 tokens break protocol invariant

Low

[L-1] TSwapPool::LiquidityAdded event is emitted with paramters out of order, causing it to log incorrect information

Description: When the `LiquidityAdded` event is emitted in the `TSwapPool::_addLiquidityMintAndTransfer` function, it logs values in an incorrect order. The `poolTokensToDeposit` value should go in the third parameter position, whereas the `wethToDeposit` value should go second.

Impact: Event emission is incorrect, leading to off-chain function potentially malfunctioning.

Recommended Mitigation:

```
1 - emit LiquidityAdded(msg.sender, poolTokensToDeposit, wethToDeposit);
2 + emit LiquidityAdded(msg.sender, wethToDeposit, poolTokensToDeposit);
```

[L-2] TSwapPool::swapExactInput doesn't specify what value to return, which makes it always return 0

Description: The `TSwapPool::swapExactInput` function is expected to return the actual amount of tokens bought by the caller. However, while it declares the named return `output`, it is never assigned a value, nor uses an explicit return statement.

Impact: The return value will always be 0, giving incorrect information to the caller.

Proof of Concept:

Include this test in `TSwapPool.t.sol`:

```
1     function testWrongOutputInSwapExactInput() public {
2         vm.startPrank(liquidityProvider);
3         weth.approve(address(pool), 100e18);
4         poolToken.approve(address(pool), 100e18);
5         pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
```

```
6         vm.stopPrank();
7
8         uint256 expectedOutput =
9             pool.getOutputAmountBasedOnInput(1e18, poolToken.balanceOf(
10                 address(pool)), weth.balanceOf(address(pool)));
11
12         vm.startPrank(user);
13         poolToken.approve(address(pool), 1e18);
14         uint256 actualOutput = pool.swapExactInput(poolToken, 1e18,
15             weth, 0, uint64(block.timestamp));
16
17         // The expected the return value from swapExactInput isn't the
18         // same as the one calculated with
19         // getOutputAmountBasedOnInput
20         assert(actualOutput != expectedOutput);
21         // swapExactInput always returns 0
22         assertEq(actualOutput, 0);
23     }
```

Recommended Mitigation:

```
1         returns (
2             uint256 output
3         )
4     {
5         uint256 inputReserves = inputToken.balanceOf(address(this));
6         uint256 outputReserves = outputToken.balanceOf(address(this));
7
8         -         uint256 outputAmount = getOutputAmountBasedOnInput(inputAmount
9             , inputReserves, outputReserves);
10        +         output = getOutputAmountBasedOnInput(inputAmount,
11             inputReserves, outputReserves);
12
13        -         if (outputAmount < minOutputAmount) {
14        -             revert TSwapPool__OutputTooLow(outputAmount,
15             minOutputAmount);
16        +         if (output < minOutputAmount) {
17        +             revert TSwapPool__OutputTooLow(output, minOutputAmount);
18        +         }
19
20        -         _swap(inputToken, inputAmount, outputToken, outputAmount);
21        +         _swap(inputToken, inputAmount, outputToken, output);
22     }
```

[L-3] TSwapPool::getPriceOfOnePoolTokenInWeth returns incorrect price for tokens that don't use 18 decimals

Description: `getPriceOfOnePoolTokenInWeth` has a hardcoded `1e18` amount to account for one token. For tokens like USDC, that has 6 decimals, this function would return an incorrect price. Tokens with decimals higher than 18 would also return incorrect prices.

Tokens like `LowDecimals.sol` and `HighDecimals.sol` from `weird-erc20` repo are simple examples of this tokens.

Impact: Pools of tokens with decimals different than 18 would get a wrong answer when calling `getPriceOfOnePoolTokenInWeth`

Recommended Mitigation:

Make the following changes to the function:

```
1  function getPriceOfOnePoolTokenInWeth() external view returns (
    uint256) {
2  +      uint8 poolTokenDecimals = ERC20(address(i_poolToken)).decimals
    ();
3  +      uint256 onePoolToken = 10**uint256(poolTokenDecimals);
4      return getOutputAmountBasedOnInput(
5  -      1e18, i_poolToken.balanceOf(address(this)), i_wethToken.
    balanceOf(address(this))
6  +      onePoolToken, i_poolToken.balanceOf(address(this)),
    i_wethToken.balanceOf(address(this))
7      );
8  }
```

Informational**[I-1] PoolFactory::PoolFactory__PoolDoesNotExist is not used and should be removed****Recommended Mitigation:**

```
1  - error PoolFactory__PoolDoesNotExist(address tokenAddress);
```

[I-2] Lacking zero address checks in PoolFactory and TSwapPool constructor**Recommended Mitigation:**

- Found in `src/PoolFactory.sol` Line: 41

```
1     constructor(address wethToken) {
2 +       if(wethToken == address(0)) {
3 +         revert();
4 +       }
5       i_wethToken = wethToken;
6     }
```

- Found in src/TSwapPool.sol Line: 67

```
1     constructor(
2       address poolToken,
3       address wethToken,
4       string memory liquidityTokenName,
5       string memory liquidityTokenSymbol
6     )
7     ERC20(liquidityTokenName, liquidityTokenSymbol)
8     {
9 +       if(poolToken == address(0) || wethToken == address(0)) {
10 +         revert();
11 +       }
12       i_wethToken = IERC20(wethToken);
13       i_poolToken = IERC20(poolToken);
14     } constructor(address tokenA, address tokenB, address factory) {
```

[I-3] PoolFactory::createPool should use .symbol() instead of .name()

Description:

Impact:

Proof of Concept:

Recommended Mitigation:

```
1 -   string memory liquidityTokenSymbol = string.concat("ts", IERC20(
2     tokenAddress).name());
2 +   string memory liquidityTokenSymbol = string.concat("ts", IERC20(
3     tokenAddress).symbol());
```

[I-4]: Event is missing indexed fields

Index event fields make the field more quickly accessible to off-chain tools that parse events. However, note that each index field costs extra gas during emission, so it's not necessarily best to index the maximum allowed per event (three fields). Each event should use three indexed fields if there are three

or more fields, and gas usage is not particularly of concern for the events in question. If there are fewer than three fields, all of the fields should be indexed.

4 Found Instances

- Found in src/PoolFactory.sol Line: 36

```
1 event PoolCreated(address tokenAddress, address poolAddress);
```

- Found in src/TSwapPool.sol Line: 43

```
1 event LiquidityAdded(address indexed liquidityProvider,  
    uint256 wethDeposited, uint256 poolTokensDeposited);
```

- Found in src/TSwapPool.sol Line: 44

```
1 event LiquidityRemoved(address indexed liquidityProvider,  
    uint256 wethWithdrawn, uint256 poolTokensWithdrawn);
```

- Found in src/TSwapPool.sol Line: 46

```
1 event Swap(address indexed swapper, IERC20 tokenIn, uint256  
    amountTokenIn, IERC20 tokenOut, uint256 amountTokenOut);
```

[I-5] Functions are missing natspec comments and should be added

Description: The functions `PoolFactory::createPool`, `TSwapPool::swapExactInput` and `TSwapPool::swapExactOutput` are missing natspec comments. These comments are important for developers to understand the purpose of the functions and how to interact with them.

Gas

[G-1] PoolFactory__PoolDoesNotExist isn't used and should be removed

Description: The `PoolFactory__PoolDoesNotExist` error is declared but never used in the codebase. This is a waste of gas.

Recommended Mitigation:

```
1 - error PoolFactory__PoolDoesNotExist(address tokenAddress);
```

[G-2] swapExactInput is not used internally and could be turned into external

Instead of marking a function as **public**, consider marking it as **external** if it is not used internally. This saves gas.

```
1      function swapExactInput(  
2          IERC20 inputToken,  
3          uint256 inputAmount,  
4          IERC20 outputToken,  
5          uint256 minOutputAmount,  
6          uint64 deadline  
7      )  
8  -      public  
9  +      external
```

[G-3] Unused variable in TSwapPool::deposit is a waste of gas

Description: The `TSwapPool::deposit` function has an unused variable `poolTokenReserves`. This variable is assigned a value but never used in the function. This is a waste of gas.

Recommended Mitigation:

```
1  - uint256 poolTokenReserves = i_poolToken.balanceOf(address(this));
```

[G-4] Unnecessary parameter in TSwapPool__WethDepositAmountTooLow error

Description: The `TSwapPool__WethDepositAmountTooLow` error includes the `MINIMUM_WETH_LIQUIDITY` parameter, which is always the same number. This wastes gas. This parameter is not necessary and should be removed.

Recommended Mitigation:

```
1  - error TSwapPool__WethDepositAmountTooLow(uint256 minimumWethDeposit,  
2      uint256 wethToDeposit);  
3  +  
4  + error TSwapPool__WethDepositAmountTooLow(uint256 wethToDeposit);  
5  +  
6  + if (wethToDeposit < MINIMUM_WETH_LIQUIDITY) {  
7  -     revert TSwapPool__WethDepositAmountTooLow(  
8  +         MINIMUM_WETH_LIQUIDITY, wethToDeposit);  
9  +     revert TSwapPool__WethDepositAmountTooLow(wethToDeposit);  
10     }  
11 }
```